154prob.mws

```
> restart:
```

Problema puntuable 27 noviembre 1996

Jose Mª Goicolea, noviembre 2000

Enunciado

En el mecanismo plano de la figura, la barra OA gira alrededor del punto fijo O con velocidad angular phi' constante. Un disco de radio R se mueve de forma que desliza sobre el eje X a la vez que rueda sin deslizar sobre la barra OA. Se pide:

- Velocidad y aceleracion del centro del disco
- Velocidad y aceleracion del centro C del disco
- Ecuacion de la polar fija referida a OXY
- Velocidad y aceleracion del punto B cuando la barra forma 60º con la horizontal



1.-

El CIR esta situado sobre la vertical por el centro del disco C, en la misma barra. Coordenadas del CIR en ejes fijos:

```
> XC:=R/tan(phi/2);
YC:=XC*tan(phi);
```

$$XC := \frac{1}{\tan\left(\frac{1}{2}\phi\right)}$$

$$YC := \frac{\tan(\phi)}{\tan\left(\frac{1}{2}\phi\right)}$$

Velocidad de C:

> **vC:=diff(subs(phi=phi(t),XC),t);**

$$vC := -\frac{1}{2}\frac{\left(1+\tan\left(\frac{1}{2}\phi(t)\right)^2\right)\left(\frac{\partial}{\partial t}\phi(t)\right)}{\tan\left(\frac{1}{2}\phi(t)\right)^2}$$

Por otra parte, se puede interpretar como una rotacion alrededor del CIR, cuya distancia a C es R/cos (phi), por lo que

> **Omega:=vC/(R/cos(phi(t)));**

$$\Omega := -\frac{1}{2}\frac{\left(1+\tan\left(\frac{1}{2}\phi(t)\right)^2\right)\left(\frac{\partial}{\partial t}\phi(t)\right)\cos(\phi(t))}{\tan\left(\frac{1}{2}\phi(t)\right)^2}$$

> **Omega:=simplify(Omega);**

$$\Omega := \frac{1}{2}\frac{\left(\frac{\partial}{\partial t}\phi(t)\right)\cos(\phi(t))}{-1+\cos\left(\frac{1}{2}\phi(t)\right)^2}$$

> **alpha:=diff(Omega,t);**

$$\alpha := \frac{1}{2}\frac{\left(\frac{\partial^2}{\partial t^2}\phi(t)\right)\cos(\phi(t))}{\left(-1+\cos\left(\frac{1}{2}\phi(t)\right)\right)^2} - \frac{1}{2}\frac{\left(\frac{\partial}{\partial t}\phi(t)\right)^2\sin(\phi(t))}{\left(-1+\cos\left(\frac{1}{2}\phi(t)\right)\right)^2} + \frac{\frac{1}{2}\left(\frac{\partial}{\partial t}\phi(t)\right)^2\cos(\phi(t))\cos\left(\frac{1}{2}\phi(t)\right)\sin\left(\frac{1}{2}\phi(t)\right)}{\left(-1+\cos\left(\frac{1}{2}\phi(t)\right)^2\right)^2}$$

Teniendo en cuenta que la velocidad de phi es constante,

```
> alpha:=simplify(subs(diff(phi(t),t,t)=0,alpha));
```

$$\alpha := -\frac{1}{2}\frac{\left(\frac{\partial}{\partial t}\phi(t)\right)^2\left(-\sin(\phi(t))+\sin(\phi(t))\cos\left(\frac{1}{2}\phi(t)\right)^2 - \cos(\phi(t))\cos\left(\frac{1}{2}\phi(t)\right)\sin\left(\frac{1}{2}\phi(t)\right)\right)}{\left(-1+\cos\left(\frac{1}{2}\phi(t)\right)^2\right)^2}$$

La expresi´on queda aqu'i un poco complicada, se puede simplificar a mano sin gran dificultad (v´ease soluci´on tipografiada en internet).

2.-

```
> aC:=diff(vC,t);
```

$$aC := \frac{1}{2}\frac{\left(1+\tan\left(\frac{1}{2}\phi(t)\right)^2\right)^2\left(\frac{\partial}{\partial t}\phi(t)\right)^2}{\tan\left(\frac{1}{2}\phi(t)\right)^3} - \frac{1}{2}\frac{\left(1+\tan\left(\frac{1}{2}\phi(t)\right)^2\right)\left(\frac{\partial}{\partial t}\phi(t)\right)^2}{\tan\left(\frac{1}{2}\phi(t)\right)} - \frac{1}{2}\frac{\left(1+\tan\left(\frac{1}{2}\phi(t)\right)^2\right)\left(\frac{\partial^2}{\partial t^2}\phi(t)\right)}{\tan\left(\frac{1}{2}\phi(t)\right)^2}$$

```
> aC:=simplify(subs(diff(phi(t),t,t)=0,aC));
```

$$aC := \frac{1}{2}\frac{\left(1+\tan\left(\frac{1}{2}\phi(t)\right)^2\right)\left(\frac{\partial}{\partial t}\phi(t)\right)^2}{\tan\left(\frac{1}{2}\phi(t)\right)^3}$$

3.-

```
> with(plots):with(plottools):
```

```
Warning, the name changecoords has been redefined
```

Dibujo de elementos del sistema, CIR y polar fija

```
> R:=1: L:=6*R:
```

```
> barra:=phi->arrow([0,0], [L*cos(phi),L*sin(phi)], L/50, L/50, .0, color=blue):
```

```
> disco:=phi->disk([R*(1+cos(phi))/sin(phi),R], R, color=yellow):
```

```
> CIR:=phi->disk([R*(1+cos(phi))/sin(phi),R*(1+cos(phi))/cos(phi)], L/50,
color=white,thickness=2):
```

```
> PF:=plot([R*(1+cos(phi))/sin(phi),R*(1+cos(phi))/cos(phi),phi=Pi/9..Pi/2.5],
thickness=4):
```

```
> display([barra(Pi/6),disco(Pi/6),CIR(Pi/6),PF],scaling=constrained);
```



Dibujo y animacion de la polar movil

Angulo girado por el disco, en funcion del angulo Phi de la barra

```
> int(-cos(phi)/(1-cos(phi)),phi);
```

$$\frac{1}{\tan\left(\frac{1}{2}\phi\right)}+2\arctan\left(\tan\left(\frac{1}{2}\phi\right)\right)$$

```
> alpha:=Phi->int(-cos(phi)/(1-cos(phi)),phi=Pi/9..Phi);
```

$$\alpha := \Phi \rightarrow \int_{\frac{1}{9}\pi}^{\Phi} -\frac{\cos(\phi)}{1-\cos(\phi)}\, d\phi$$

Coordenadas del centro del disco, en funcion de Phi

```
> XO:=Phi->R*(1+cos(Phi))/sin(Phi);
YO:=Phi->R;
```

$$XO := \Phi \rightarrow \frac{R\,(1+\cos(\Phi))}{\sin(\Phi)}$$

$$YO := \Phi \rightarrow R$$

coordenadas del CIR en ejes moviles

```
> xC:=Phi->(R*(1+cos(Phi))/cos(Phi)-R)*sin(alpha(Phi));
yC:=Phi->(R*(1+cos(Phi))/cos(Phi)-R)*cos(alpha(Phi));
```

$$xC := \Phi \rightarrow \left(\frac{R\,(1+\cos(\Phi))}{\cos(\Phi)} - R\right)\sin(\alpha(\Phi))$$

$$yC := \Phi \rightarrow \left(\frac{R\,(1+\cos(\Phi))}{\cos(\Phi)} - R\right)\cos(\alpha(\Phi))$$

Polar movil en ejes moviles

```
> plot([xC(Phi),yC(Phi),Phi=Pi/9..Pi/2.5],scaling=constrained,thickness=3);
```



parametrizacion del angulo girado en funcion de variable p entera

> q:=p->Pi/9+p*(Pi/200);

$$q := p \to \frac{1}{9}\pi + \frac{1}{200}p\,\pi$$

dibujo de un radio del disco que gira con el mismo

```
> radio:=Phi->line([XO(Phi),YO(Phi)],
[XO(Phi)+R*cos(alpha(Phi)),YO(Phi)+R*sin(alpha(Phi))], color=brown, linestyle=3,
thickness=3):
```

```
> anim:=seq(display(PF,
barra(q(p)),disco(q(p)),radio(q(p)),CIR(q(p)),
plot([XO(q(p))+xC(Phi)*cos(alpha(q(p)))-yC(Phi)*sin(alpha(q(p))),
YO(q(p))+xC(Phi)*sin(alpha(q(p)))+yC(Phi)*cos(alpha(q(p))),
Phi=Pi/9..Pi/2.5],thickness=3,color=green)
),p=0..55):
```

```
> display(anim,insequence=true,scaling=constrained);
```